



Les nombres complexes

© Pierre Lantagne

Enseignant retraité du Collège de Maisonneuve

La première version de ce document est parue en septembre 2000. Le but de ce document est double. D'une part, transposer en Maple certains éléments de connaissance sur les nombres complexes vus en classe. D'autre part, faire pleinement réaliser à l'élève que le référentiel utilisé par Maple pour la résolution d'équations est l'ensemble des nombres complexes. Dans ce document, plusieurs exemples sont développés afin de montrer la nécessité d'être circonspect dans l'utilisation de Maple pour la résolution d'équations et même pour le tracé de graphiques cartésiens de fonctions réelles à valeurs réelles. On retrouvera dans ce document quelques macro-commandes maisons qui ont été développées pour contrer (dans une démarche pédagogique) certains effets non désirables du mécanisme de la simplification automatique de Maple.

Bonne lecture à tous !

* Ce document Maple est exécutable avec la version 2020.1

Initialisation

```
> restart;  
> with(plottools,arrow,disk,line):  
  with(plots,display,implicitplot,textplot,setoptions):  
> setoptions(xtickmarks=12,  
             ytickmarks=12,  
             labels=["Re","Im"],  
             axesfont=[axes,"TIMES",8],  
             labeldirections=[horizontal,vertical],  
             size=[300,300]):  
> interface(showassumed=0):
```

L'initialisation suivante permettra d'avoir plus de lisibilité des nombres décimaux en supprimant les zéros non significatifs à la fin d'un nombre.

```
> interface(typesetting=extended); # Pour s'assurer le niveau de  
  composition étendue  
Typesetting:-Settings(striptrailing=true);  
                                extended  
                                false
```

(1.1)

Calculs arithmétiques dans \mathbb{C}

Les requêtes suivantes montreront qu'effectivement la simplification automatique de Maple est effectuée dans le corps des nombres complexes \mathbb{C} .

Dans Maple, l'unité imaginaire est noté par défaut par la lettre I majuscule au lieu de la lettre i minuscule comme nous l'avons vu en classe.

```
> I^2;
```

$$\begin{array}{|l} \hline \\ \hline \end{array} \quad -1 \quad (2.1)$$

$$\begin{array}{|l} \hline > 1/I; \\ \hline \end{array} \quad -I \quad (2.2)$$

$$\begin{array}{|l} \hline > ((1-I)*(2+I)) / ((3+I)/(4-I)); \\ \hline \end{array} \quad \frac{13}{5} - \frac{16I}{5} \quad (2.3)$$

$$\begin{array}{|l} \hline > (2-I)^5; \\ \hline \end{array} \quad -38 - 41 I \quad (2.4)$$

$$\begin{array}{|l} \hline > f:=x->2*x^2-I*x+1; \\ > 'f'(2+3*I)=f(2+3*I); \\ \hline \end{array} \quad \begin{array}{l} f := x \mapsto 2 \cdot x^2 - I \cdot x + 1 \\ f(2 + 3I) = -6 + 22 I \end{array} \quad (2.5)$$

L'utilisateur a la possibilité de changer la notation de l'unité imaginaire par une lettre de son choix: par la lettre i ou la lettre j entre autres.

$$\begin{array}{|l} \hline > interface(imaginaryunit=i); \\ \hline \end{array} \quad I \quad (2.6)$$

$$\begin{array}{|l} \hline > i^2; \\ \hline \end{array} \quad -1 \quad (2.7)$$

$$\begin{array}{|l} \hline > 1/i; \\ \hline \end{array} \quad -i \quad (2.8)$$

$$\begin{array}{|l} \hline > ((1-i)*(2+i)) / ((3+i)/(4-i)); \\ \hline \end{array} \quad \frac{13}{5} - \frac{16i}{5} \quad (2.9)$$

$$\begin{array}{|l} \hline > (2-i)^5; \\ \hline \end{array} \quad -38 - 41 i \quad (2.10)$$

$$\begin{array}{|l} \hline > f:=x->2*x^2-i*x+1; \\ > 'f'(2+3*i)=f(2+3*i); \\ \hline \end{array} \quad \begin{array}{l} f := x \mapsto 2 \cdot x^2 - i \cdot x + 1 \\ f(2 + 3i) = -6 + 22 i \end{array} \quad (2.11)$$

Pour revenir au choix par défaut de Maple, il faudrait saisir la requête `interface(imaginaryunit=I)` tout simplement. Pour la suite de ce document, nous allons tout de même conserver la lettre i minuscule.

Maple effectue la résolution des équations dans \mathbb{C} .

$$\begin{array}{|l} \hline > \text{Éq}:=i*z+4=3*z-i; \\ > \text{'E.-S.'}=solve(\text{Éq},\{z\}); \\ \hline \end{array} \quad \begin{array}{l} \text{Éq} := iz + 4 = 3z - i \\ E.-S. = \left\{ z = \frac{11}{10} + \frac{7i}{10} \right\} \end{array} \quad (2.12)$$

```
> Éq:=x^2*(x^2-3*x+9);
`E.-S.`=solve(Éq,x);
```

$$\begin{aligned} \text{Éq} &:= x^2 (x^2 - 3x + 9) \\ E.-S. &= \left(0, 0, \frac{3}{2} + \frac{3i\sqrt{3}}{2}, \frac{3}{2} - \frac{3i\sqrt{3}}{2} \right) \end{aligned} \quad (2.13)$$

C'est de cette façon que Maple signifie que 0 est une racine double de cette équation.

```
> Éq:=x^4+4=0;
`E.-S.`=solve(Éq,x);
```

$$\begin{aligned} \text{Éq} &:= x^4 + 4 = 0 \\ E.-S. &= (1 - i, 1 + i, -1 - i, -1 + i) \end{aligned} \quad (2.14)$$

```
> Éq:=x^6-6*x^5+10*x^4-x^2+6*x-10=0;
`E.-S.`=solve(Éq,x);
```

$$\begin{aligned} \text{Éq} &:= x^6 - 6x^5 + 10x^4 - x^2 + 6x - 10 = 0 \\ E.-S. &= (1, -1, i, -i, 3 - i, 3 + i) \end{aligned} \quad (2.15)$$

La macro-commande `conjugate()` permet d'obtenir le conjugué d'un nombre complexe z .

```
> z:=a+b*i;
conjugate(z);
```

$$\begin{aligned} z &:= a + ib \\ \overline{(a + ib)} & \end{aligned} \quad (2.16)$$

Le résultat précédent montre clairement que Maple considère par défaut les lettres a et b comme des constantes complexes quelconques.

Considérons alors $a, b \in \mathbb{R}$. Ainsi

```
> assume(a,real,b,real);
> 'conjugate'(z)=conjugate(z);
```

$$\overline{(a + ib)} = a - ib \quad (2.17)$$

Libérons les variables a et b .

```
> a:='a':
b:='b':
```

Propriétés des conjugués

Soit $z = a + bi$ ($a, b \in \mathbb{R}$), z_1 et z_2 trois nombres complexes quelconques. Alors

Propriété 1: $a^2 + b^2$ se développe dans \mathbb{C} par $a^2 + b^2 = z \overline{z}$. (C'est le développement algébrique de la **somme de deux carrés**.)

```
> assume(a,real,b,real);
z:=a+b*i;
```

$$z := a + ib \quad (3.1)$$

```
> 'z*conjugate(z)'=expand(z*conjugate(z));
```

$$z \overline{z} = a^2 + b^2 \quad (3.2)$$

Propriété 2: $z = \overline{z}$ si et seulement si $\Im(z) = 0$.

En effet, lorsque $b = 0$,

```
> b:=0: # a est toujours réel n'ayant pas été rendu libre
z:=a+b*i;
'conjugate'(z)=conjugate(z);
a:='a':b:='b': # Libérons a et b
z := a
a = a
```

(3.3)

Propriété 3: Le conjugué du conjugué de z est z .

```
> z:='z': # libérons z
'conjugate'('conjugate'(z))=conjugate(conjugate(z));
z = z
```

(3.4)

Propriété 4: Le conjugué d'une somme est la somme des conjugués.

```
> conjugate(z[1]+z[2])=expand(conjugate(z[1]+z[2]));
(z1+z2) = z1 + z2
```

(3.5)

Propriété 5:

```
> conjugate(z[1]-z[2])=expand(conjugate(z[1]-z[2]));
(z1-z2) = z1 - z2
```

(3.6)

Propriété 6: Le conjugué d'un produit est le produit des conjugués.

```
> conjugate(z[1]*z[2])=expand(conjugate(z[1]*z[2]));
(z1*z2) = z1*z2
```

(3.7)

Propriété 7: Le conjugué d'un quotient est le quotient des conjugués si le dénominateur est non nul.

Si $z \neq 0$,

```
> conjugate(z[1]/z[2])=expand(conjugate(z[1]/z[2]));
(z1/z2) = z1/z2
```

(3.8)

Représentation graphique des nombres complexes

La représentation graphique des nombres complexes est souvent appelée diagramme d'Argand ou plan de Gauss en l'honneur des mathématiciens qui ont utilisé cette représentation dans leurs travaux, Jean Robert Argand (suisse, 1768-1822) en 1806 et Karl Friedrich Gauss (allemand 1766-1855) en 1831. On dit habituellement plan complexe.

Un plan complexe est en fait un plan cartésien où l'abscisse est renommée l'axe de la partie réelle et l'ordonnée renommée axe de la partie imaginaire.

Réprésentons dans le plan complexe le nombre $z = 3 + 2i$.


```
> z:=3+2*i:
Point:=disk([Re(z),Im(z)],0.025,color=navy):
P1:=textplot([Re(z)+0.05,Im(z),P(z)],font=[TIMES,8],color=navy,align=
```

```
{RIGHT}):
```

```
LH:=line([0, Im(z)],[Re(z),Im(z)],color=grey,linestyle=dot):
```

```
LV:=line([Re(z),0],[Re(z),Im(z)],color=grey,linestyle=dot):
```

```
display({Point,P1,LH,LV},view=[-1..4,-1..3]);
```



Tout comme dans un plan cartésien, un point dans un plan complexe est bien une entité géométrique mais la notion de coordonnées d'un point est une notion qui n'est pas définie dans un plan complexe.

Le point P de ce plan complexe étant la représentation géométrique du nombre $z = 3 + 2i$, on dit que $z = 3 + 2i$ est l'afixe du point P ou que le point P a pour affixe $3 + 2i$.

Notons que

- l'opposé de $z = a + bi$ ($a, b \in \mathbb{R}$), soit $-z$, est représenté par le point symétrique par rapport à l'origine du point d'affixe z .
- le conjugué de $z = a + bi$, soit \overline{z} , est représenté par le point symétrique par rapport à l'axe des réels du point d'affixe z .

```
> z:=5+2*i:
```

```
Point:=disk([Re(z),Im(z)],0.075,color=navy):
```

```
LHPoint:=line([0, Im(z)],[Re(z),Im(z)],color=grey,linestyle=dot):
```

```
LVPoint:=line([Re(z),0],[Re(z),Im(z)],color=grey,linestyle=dot):
```

```
T1:=textplot([Re(z)+0.075,Im(z),P(z)],font=[TIMES,10],color=navy,  
align={above,right}):
```


```
Point1:=disk([-Re(z),-Im(z)],0.075,color=navy):
```

```
LHPoint1:=line([0,-Im(z)],[-Re(z),-Im(z)],color=grey,linestyle=dot):
```


```
LVPoint1:=line([-Re(z),0],[-Re(z),-Im(z)],color=grey,linestyle=dot):
```

```
T2:=textplot([-Re(z)+0.05,-Im(z),Q(-z)],font=[TIMES,10],color=navy,  
align={below,left}):
```

```
display({Point,Point1,T1,T2,LHPoint,LVPoint,LHPoint1,LVPoint1},view=  
[-10..10,-5..5]);
```



```
> z:=5+2*i:
Point:=disk([Re(z),Im(z)],0.075,color=navy):
LHPoint:=line([0, Im(z)],[Re(z),Im(z)],color=grey,linestyle=dot):
LVPoint:=line([Re(z),0],[Re(z),Im(z)],color=grey,linestyle=dot):
T1:=textplot([Re(z)+0.05,Im(z),P(z)], font=[TIMES,8],color=navy,
align={RIGHT}):
Point1:=disk([Re(conjugate(z)),Im(conjugate(z))],0.075,color=navy):
LHPoint1:=line([0,-Im(z)],[Re(z),-Im(z)],color=grey,linestyle=dot):
LVPoint1:=line([Re(z),0],[Re(z),-Im(z)],color=grey,linestyle=dot):
T2:=textplot([Re(conjugate(z))+0.05,Im(conjugate(z)),Q(conjugate(z))
],font=[TIMES,8],color=navy,align={RIGHT}):
display({Point,Point1,T1,T2,LHPoint,LVPoint,LHPoint1,LVPoint1},view=
[-2..7,-3..3]);
```



Dans quel quadrant sera situé le point dont l'affixe est le conjugué de l'opposé de z , soit $-\overline{z}$?

Et celui dont l'affixe est l'opposé du conjugué de z , soit $-\overline{z}$?

Votre réponse:

Vérification.


Votre réponse:

Vérification.

Situons dans un plan complexe $i^2, i^3, i^4, i^5, i^6, i^7$ et i^8 .

```
> Nombres:=seq(i^k,k=1..8);
      Nombres := [i, -1, -i, 1, i, -1, -i, 1] (4.1)
```

```
> Points:=seq(disk([Re(Nombres[k]),Im(Nombres[k])],0.04,color=black),k=
1..8):
display(Points,view=[-2..2,-2..2]);
```




Remarquons que la représentation de ces huit affixes ne donnent dans le plan complexe que seulement quatre points distincts. Pourquoi ?

Même question pour les huit affixes suivants: $i^{-2}, i^{-3}, i^{-4}, i^{-5}, i^{-6}, i^{-7}$ et i^{-8} .

```
> Nombres:=seq(i^(-k),k=1..8);
      Nombres := [-i, -1, i, 1, -i, -1, i, 1] (4.2)
```

```
> Points:=seq(disk([Re(Nombres[k]),Im(Nombres[k])],0.04,color=black),k=
1..8):
display(Points,view=[-2..2,-2..2]);
```




Une autre façon de représenter graphiquement un nombre complexe $z = a + bi$ ($a, b \in \mathbb{R}$) dans un plan complexe est de le représenter par un vecteur géométrique issu de l'origine O et se terminant au point A d'affixe z .

```
> z:= -4+2*i;
      z := -4 + 2i (4.3)
```

```
> V:= arrow([0,0], [Re(z),Im(z)], .1, .3, .10,color=red):
S:=textplot([Re(z),Im(z),'A '(z)],font=[TIMES,10],align={above,right})
```

```
);
display({S,V},view=[-6..3,-3..3]);
```



Le module d'un nombre complexe

Le module d'un nombre complexe z , noté $|z|$, est définie par $|z| = \sqrt{z \overline{z}}$.

En conséquence, si $z = a + bi$ ($a, b \in \mathbb{R}$), $|z| = \sqrt{a^2 + b^2}$.

Par sa notation même, on dit que le module d'un nombre complexe est le prolongement dans \mathbb{C} de la notion de la valeur absolue dans \mathbb{R} .

- La **valeur absolue** d'un nombre réel exprime la distance, sur la droite réelle, entre l'origine et le point représentant ce nombre réel.
- Similairement, le **module** d'un nombre complexe exprimera dans un plan complexe
 - soit la distance entre l'origine et le point d'affixe z ,
 - soit la longueur du vecteur représentant ce nombre.

Les développements suivants donneront l'occasion au lecteur d'explorer des lieux géométriques dans un plan complexe dont la description est donnée à l'aide de la partie réelle d'un nombre z , soit $\Re(z)$, de la partie imaginaire de z , soit $\Im(z)$ ou soit avec le module de z , noté $|z|$.

Soit $z = x + yi$ ($x, y \in \mathbb{R}$), un nombre complexe quelconque sous la forme rectangulaire (forme cartésienne).


Quel est le lieu géométrique dans un plan complexe de tous les points pour lesquels nous avons $\Re(z) = \Im(z)$?

```
> assume(x,real,y,real);
z:=x+y*i;
```

$$z := x + iy$$

(5.1)


```
> Lieu:=implicitplot(Re(z)=Im(z),Re(z)=-3..3,Im(z)=-3..5,
color=navy);
> display(Lieu);
```

Le lieu géométrique est une droite passant par l'origine ayant une orientation de 45 degrés, c'est-à-dire de pente 1.

Quel est le lieu géométrique dans le plan complexe de tous les points pour lesquels nous avons $1 \leq |z| < 5$?

```
> A:=implicitplot(abs(z)-5,Re(z)=-5..5,Im(z)=-5..5,
    coloring=["orange","yellow"],
    filledregions=true,
    linestyle=2,thickness=2):
B:=implicitplot(abs(z)-1,Re(z)=-1..1,Im(z)=-1..1,
    coloring=["yellow","orange"],
    filledregions=true,
    thickness=2):
display([B,A]);
```



C'est une couronne centrée à l'origine comprise entre un disque de rayon 1 et un disque de rayon 5, incluant la circonférence du disque de rayon 1 et excluant celle de rayon 5.

Libérons les variables x et y .

```
> x:='x':
y:='y':
```

Soit maintenant $z = a + bi$ et $w = c + di$ ($a, b, c, d \in \mathbb{R}$). Développons $|z - w|$.

```
> z:=a+b*i:
w:=c+d*i:
assume(a,real,b,real,c,real,d,real);
'abs(z-w)'=abs(z-w);
```

$$|z - w| = \sqrt{(a - c)^2 + (b - d)^2}$$

(5.2)

Rappelons la formule de la distance entre deux points dans un plan cartésien.

```
> d:=(P,Q)->sqr((P[1]-Q[1])^2+(P[2]-Q[2])^2);
A:=[a,b]:
B:=[c,d]:
'd(A,B)'=d(A,B);
```

$$d := (P, Q) \mapsto \sqrt{(P_1 - Q_1)^2 + (P_2 - Q_2)^2}$$

$$d(A, B) = \sqrt{(a - c)^2 + (b - d)^2}$$


(5.3)

Ainsi $|z - w|$ correspond géométriquement à la distance, dans un plan complexe, entre les points d'affixes z et w .

Il en est de même évidemment pour $|w - z|$.

Quel est le lieu des points d'affixe z pour lesquels $|z - 2i|$?


```
> Lieu:=implicitplot(abs(z-2*i)=2,Re(z)=-2..2,Im(z)=-1..4):
> display(Lieu,scaling=constrained);
```



Le lieu géométrique des points d'affixe z satisfaisant l'équation $|z - 2i| = 2$ est la circonférence d'un cercle de rayon 2 centré au point d'affixe $2i$.

Qu'en est-il du lieu d'inéquation $|z + 2i| \leq \frac{3}{4}$?

```
> Lieu:=implicitplot(abs(z+2*i)-3/4,Re(z)=-2..2,Im(z)=-3..1,
    gridrefine=2,coloring=["orange","yellow"],filledregions=true):
> display(Lieu);
```




C'est donc un disque centré au point d'affixe $-2i$ de rayon $\frac{3}{4}$ incluant la circonférence.

Qu'en est-il du lieu d'inéquations $1 < |z - 4 + i| < 4$?

C'est une couronne comprise entre deux disques de rayon 4 et 1 centrée au point d'affixe $4 - i$. Les circonférences respectives étant exclues.

```
> A:=implicitplot(abs(z-4+i)-4,Re(z)=-2..9,Im(z)=-6..4,
    coloring=["orange","yellow"],filledregions=true,
    linestyle=2,thickness=2):
B:=implicitplot(abs(z-4+i)-1,Re(z)=3..5,Im(z)=-2..0,
    coloring=["yellow","orange"],filledregions=true,
    linestyle=2,thickness=2,gridrefine=2):
> display([B,A]);
```



Libérons z .

```
> z:='z';
```

$z := z$

(5.4)

Propriétés des modules

Soit z , z_1 et z_2 trois nombres complexes quelconques.

Propriété 1: Le module d'un nombre complexe est non négatif.

```
> is(abs(z)>=0);
```

$true$

(6.1)

Propriété 2: Le module d'un nombre complexe est égal au module de son conjugué.

```
> 'abs'(conjugate(z))=expand('abs'(conjugate(z)));
evalb(%);
```

$|\bar{z}| = |z|$

$true$

(6.2)

Propriété 3: Le module d'un produit est égal au produit des modules.

```
> abs(z[1]*z[2])=expand(abs(z[1]*z[2]));
```

(6.3)

$$|z_1 z_2| = |z_1| |z_2| \quad (6.3)$$

Propriété 4: Le module d'un quotient est égal au quotient des modules si le dénominateur est non nul.
Si $z_2 \neq 0$,

$$\begin{aligned} & \text{> } \text{abs}(z[1]/z[2]) = \text{expand}(\text{abs}(z[1]/z[2])); \\ & \quad \left| \frac{z_1}{z_2} \right| = \frac{|z_1|}{|z_2|} \end{aligned} \quad (6.4)$$

Propriété 5: L'inégalité du triangle $|z_1 + z_2| \leq |z_1| + |z_2|$.

$$\begin{aligned} & \text{> } (\text{abs}(z[1]+z[2]) \leq \text{abs}(z[1]) + \text{abs}(z[2])); \quad \# \text{ sans commentaire !} \\ & \quad \text{evalb(\%);} \\ & \quad |z_1 + z_2| \leq |z_1| + |z_2| \\ & \quad |z_1 + z_2| \leq |z_1| + |z_2| \end{aligned} \quad (6.5)$$

Forme polaire d'un nombre complexe

Rappelons que, dans un plan polaire, un même point possède une infinité de coordonnées polaires (r, θ) , ($r, \theta \in \mathbb{R}$):

- r étant appelé la coordonnée radiale de (r, θ)
- θ étant appelé la coordonnée angulaire de (r, θ) .

Ainsi, les coordonnées polaires suivantes sont celles d'un même point dans un plan polaire.

$$\left(-\frac{1}{4}, -\frac{\pi}{3}\right), \left(-\frac{1}{4}, -\frac{\pi}{3} + 2k\pi\right), \left(\frac{1}{4}, \frac{2\pi}{3}\right), \left(\frac{1}{4}, \frac{2\pi}{3} + 2k\pi\right), k \in \mathbb{Z}$$

Comme il est possible de lier un plan cartésien à un plan complexe pour représenter les nombres complexes sous forme rectangulaire $a + bi$, on lie un plan polaire à un plan complexe également pour représenter les nombres complexes sous leur forme polaire (r, θ) , ($r, \theta \in \mathbb{R}$).

Soit un nombre complexe $z = (r, \theta)$, ($r, \theta \in \mathbb{R}$) quelconque.

Un nombre complexe écrit sous la forme polaire (r, θ) ($r, \theta \in \mathbb{R}$), est transposé en Maple par `polar(r, theta)`.

$$\begin{aligned} & \text{> } z := \text{polar}(r, \theta); \\ & \quad z := \text{polar}(r, \theta) \end{aligned} \quad (7.1)$$

Puisqu'un même point d'un plan complexe possède une infinité d'affixes polaires, l'argument θ d'un nombre complexe n'est donc pas unique. Par contre, nous savons que si $r > 0$, θ est unique modulo 2π . En clair, cela signifie que, pour deux arguments θ_1 et θ_2 distincts d'un nombre complexe sous forme polaire, si

$$\theta_2 - \theta_1 = 2k\pi \text{ alors } (r, \theta_1) = (r, \theta_2).$$

Afin d'avoir une écriture unique d'un nombre complexe sous forme polaire, on convient de choisir θ dans l'intervalle $]-\pi, \pi]$, c'est-à-dire $-\pi < \theta \leq \pi$

Ce choix unique θ est appelé argument principal.

Si $r = 0$, toute valeur de θ détermine le même nombre complexe. Dans ce cas, on convient de choisir $\theta = 0$.

Ainsi, avec Maple,

- r , le module d'un nombre complexe sous forme polaire est donné par $r = |z|$.
- θ , l'argument principal d'un nombre complexe sous forme polaire est donné par `argument(z)`.

```
> z:=polar(1/4,Pi/3);
'abs(z)'=abs(z);
'argument(z)'=argument(z);
```

$$z := \text{polar}\left(\frac{1}{4}, \frac{\pi}{3}\right)$$

$$|z| = \frac{1}{4}$$

$$\arg(z) = \frac{\pi}{3} \quad (7.2)$$

```
> z:=polar(1/4,-Pi/3);
'abs(z)'=abs(z);
'argument(z)'=argument(z);
```

$$z := \text{polar}\left(\frac{1}{4}, -\frac{\pi}{3}\right)$$

$$|z| = \frac{1}{4}$$

$$\arg(z) = -\frac{\pi}{3} \quad (7.3)$$

```
> z:=polar(1/4,-17*Pi/3);
'abs(z)'=abs(z);
'argument(z)'=argument(z);
```

$$z := \text{polar}\left(\frac{1}{4}, -\frac{17\pi}{3}\right)$$

$$|z| = \frac{1}{4}$$

$$\arg(z) = \frac{\pi}{3} \quad (7.4)$$

Donné un nombre complexe sous forme polaire $\text{polar}(r, \theta)$, le nombre r peut être négatif. Dans ce cas, l'argument θ est remplacé dans les calculs par $\theta + \pi$ si $\theta < 0$ et par $\theta - \pi$ si $\theta > 0$. Cela correspond à une rotation de π radians dans le sens anti-horaire dans le premier cas et dans le sens horaire dans le second cas.

```
> z:=polar(-1/4,2*Pi/3);
```

```
'abs(z)'=abs(z);
'argument(z)'=argument(z);
```

$$z := \text{polar}\left(-\frac{1}{4}, \frac{2\pi}{3}\right)$$

$$|z| = \frac{1}{4}$$

$$\arg(z) = -\frac{\pi}{3} \quad (7.5)$$

```
> z:=polar(-1/4,-5*Pi/3);
'abs(z)'=abs(z);
'argument(z)'=argument(z)
```

$$z := \text{polar}\left(-\frac{1}{4}, -\frac{5\pi}{3}\right)$$

$$|z| = \frac{1}{4}$$

$$\arg(z) = -\frac{2\pi}{3} \quad (7.6)$$

```
> z:='z';
```

$$z := z \quad (7.7)$$

La forme trigonométrique $r \text{cis}(\theta)$ d'un nombre complexe donné sous la forme polaire est directement obtenue avec `evalc()`.

```
> z=evalc(polar(r,theta));
```

$$z = r \cos(\theta) + i r \sin(\theta) \quad (7.8)$$

```
> z := polar(3,Pi/5);
``=evalc(z);
```

$$z := \text{polar}\left(3, \frac{\pi}{5}\right)$$

$$= 3 \cos\left(\frac{\pi}{5}\right) + 3i \sin\left(\frac{\pi}{5}\right) \quad (7.9)$$

Remarque: Si θ est un angle remarquable, le mécanisme de la simplification automatique de Maple réduit automatiquement le sinus et le cosinus.

La forme polaire d'un nombre complexe donné avec l'écriture rectangulaire est obtenue avec `convert(, polar)`.

```
> z := polar(sqrt(2),Pi/3);
``=evalc(z);
```

$$z := \text{polar}\left(\sqrt{2}, \frac{\pi}{3}\right)$$

(7.10)

$$= \frac{\sqrt{2}}{2} + \frac{i\sqrt{2}\sqrt{3}}{2} \quad (7.10)$$

```
> combine((7.10), radical);
```

$$= \frac{\sqrt{2}}{2} + \frac{i\sqrt{6}}{2} \quad (7.11)$$

La forme polaire d'un nombre complexe donné sous la forme rectangulaire est directement obtenue avec la macro-commande `convert(, polar)`.

Obtenons $1 + \sqrt{3}i$ sous la forme polaire.

```
> z := 1+sqrt(3)*i;
z:=convert(z,polar);
```

$$z := 1 + i\sqrt{3}$$

$$1 + i\sqrt{3} = \text{polar}\left(2, \frac{\pi}{3}\right) \quad (7.12)$$

Détaillons ici pour voir comment Maple s'y prend. Il s'agit en Maple de la conversion suivante:

```
> z:=a+b*i;
z:=convert(z,polar);
```

$$z := a + ib$$

$$a + ib = \text{polar}\left(\sqrt{a^2 + b^2}, \arg(a + ib)\right) \quad (7.13)$$

Où $r = |a + bi| = \sqrt{a^2 + b^2}$ est le module de z (un nombre réel positif) et $\theta = \arg(a + bi)$.

La manière de choisir $\theta = \arg(a + bi)$ est comme suit: avec $r > 0$,

- $\theta \geq 0$ si $b \geq 0$, autrement $\theta < 0$.
- pour le cas où $b = 0$, $\theta = 0$ si $a \geq 0$. Autrement $\theta = \pi$.

Bref,

- pour les nombres réels positifs (≥ 0), soit les nombres complexes z dont $\Re(z) \geq 0$ et $\Im(z) = 0$, $\arg(z) = 0$

```
> z:=0:
'argument'(z)=argument(z);
z:=1/2:
'argument'(z)=argument(z);
z:=sqrt(3):
'argument'(z)=argument(z);
```

$$\arg(0) = 0$$

$$\arg\left(\frac{1}{2}\right) = 0$$

$$\arg(\sqrt{3}) = 0 \quad (7.14)$$

- pour les nombres réels négatifs (< 0), soit les nombres complexes z dont $\Re(z) < 0$ et $\Im(z) = 0$, $\arg(z) = \pi$.

```

> z:=-1:
'argument'(z)=argument(z);
z:=-1/4:
'argument'(z)=argument(z);
z:=-sqrt(5):
'argument'(z)=argument(z);

```

$$\arg(-1) = \pi$$

$$\arg\left(-\frac{1}{4}\right) = \pi$$

$$\arg(-\sqrt{5}) = \pi \quad (7.15)$$

— pour tous les nombres imaginaires, voici comment Maple détermine l'argument d'un nombre complexe.

```

> z:=polar(a+b*i);
map(evalc, z);

```

$$z := \text{polar}\left(\sqrt{a^2 + b^2}, \arg(a + ib)\right)$$

$$\text{polar}\left(\sqrt{a^2 + b^2}, \arctan(b, a)\right)$$

(7.16)

où $\theta = \arctan(b, a) \in]-\pi, \pi]$ est l'argument principal de z .

Comme vu en classe, pour déterminer θ , l'argument principal de z , il y a plusieurs sous-cas à traiter selon les signes de la partie réelle (a) et de la partie imaginaire (b). Cependant, de nombreux langages de programmation disposent d'une variante de la fonction arctangente, souvent appelée $\text{atan2}(b, a)$ permettant le traitement de tous ces cas.

Aussi, il existe une formule utilisant la fonction arccos qui réduit de beaucoup le nombre de cas à considérer.

Sans entrer dans les détails d'un cours de mathématique, voyons quelques exemples d'argument d'un nombre imaginaire.

```

> z:=-sqrt(2)*i:
'argument'(z)=argument(z);

```

$$\arg(-i\sqrt{2}) = -\frac{\pi}{2}$$

(7.17)

```

> z:=-sqrt(2)-sqrt(2)*i:
'argument'(z)=argument(z);

```

$$\arg(-\sqrt{2} - i\sqrt{2}) = -\frac{3\pi}{4}$$

(7.18)

```

> z:=-1-sqrt(3)*i:
'argument'(z)=argument(z);

```

$$\arg(-1 - i\sqrt{3}) = -\frac{2\pi}{3}$$

(7.19)


```
> z:=3+4*i:
'argument'(z)=argument(z);
```

$$\arg(3 + 4i) = \arctan\left(\frac{4}{3}\right) \quad (7.20)$$

```
> z:=-3+4*i:
'argument'(z)=argument(z);
```

$$\arg(-3 + 4i) = -\arctan\left(\frac{4}{3}\right) + \pi \quad (7.21)$$

```
> z:=-3-4*i:
'argument'(z)=argument(z);
```

$$\arg(-3 - 4i) = \arctan\left(\frac{4}{3}\right) - \pi \quad (7.22)$$

```
> z:=polar(-1,-Pi/4):
'argument'(z)=argument(z);
```

$$\arg\left(\text{polar}\left(-1, -\frac{\pi}{4}\right)\right) = \frac{3\pi}{4} \quad (7.23)$$

Forme exponentielle $r e^{i\theta}$ d'un nombre complexe

Montrons comment intervient le mécanisme de la simplification automatique de Maple sur le nombre $4 e^{i\frac{\pi}{3}}$.

```
> z:=-4*exp(i*Pi/3);
```

$$z := -2 - 2i\sqrt{3} \quad (8.1)$$

```
> 'abs'(z)=abs(z);
'argument'(z)=argument(z);
```

$$|-2 - 2i\sqrt{3}| = 4$$

$$\arg(-2 - 2i\sqrt{3}) = -\frac{2\pi}{3} \quad (8.2)$$

Maple reconnaît bien la forme exponentielle d'un nombre complexe (heureusement !).

Soit maintenant $z = -1 - i \in \mathbb{C}$. Observons le résultat de la macro-commande `convert(, exp)` sur ce nombre.

```
> z:=-1-i:
z:=convert(z,exp);
```

$$-1 - i = -1 - i \quad (8.3)$$

Il ne faut pas être surpris ici de ce résultat (mécanisme de la simplification automatique oblige !). L'aide sur `convert/exp` est clair: cette macro-commande n'est pas du tout appropriée. Maple a ici tout simplement

transformé $\sqrt{2} e^{-\frac{3i}{4}\pi}$ automatiquement en écriture rectangulaire.

```
> sqrt(2)*exp(-(1/4)*(3*i)*Pi);
```

$$\sqrt{2} \left(-\frac{\sqrt{2}}{2} - \frac{i\sqrt{2}}{2} \right) \quad (8.4)$$

```
> simplify((8.4));
```

$$-1 - i \quad (8.5)$$

Pour contourner le mécanisme de la simplification automatique, créons notre propre macro-commande de conversion `convert(, Exp)`. (Notez le E majuscule).

```
> `convert/Exp`:= proc(valeur)
    local Nombre;
    Nombre:=convert(valeur,polar);
    if type(Nombre,function) and op(0,Nombre)='polar' then
        op(1,Nombre)*'exp'(i*op(2,Nombre))
    elif has({op(Nombre)},'polar') then map(`convert/Exp`,Nombre)
    else Nombre
    fi
end proc;
```

```
> z:=convert(z,Exp);
```

$$-1 - i = \sqrt{2} e^{-\frac{3i}{4}\pi} \quad (8.6)$$

Cette courte procédure est un exemple de possibilité de programmation qu'offre Maple. À moins d'avoir eu une introduction à la programmation Maple, il ne faut pas s'attarder à comprendre le détail de cette procédure. Ce n'est pas le but ici. Le but visé est plutôt de vous sensibiliser à ce mécanisme. Ce n'est pas toujours désiré dans une démarche pédagogique.

Effectuons quelques calculs simples.

```
> z1:=convert(-1+i,Exp);
z2:=convert(-5,Exp);
```

$$\begin{aligned} z1 &:= \sqrt{2} e^{\frac{3i}{4}\pi} \\ z2 &:= 5 e^{i\pi} \end{aligned} \quad (8.7)$$

Multiplions maintenant ces deux nombres qui sont sous la forme polaire.

```
> z1*z2;
```

$$-5\sqrt{2} \left(-\frac{\sqrt{2}}{2} + \frac{i\sqrt{2}}{2} \right) \quad (8.8)$$

Ce résultat est dû au mécanisme de la simplification automatique de Maple: pas possible de le contourner sans la macro-commande maison `convert(, Exp)`.

```
> convert(z1*z2,Exp);
```

$$5\sqrt{2} e^{-\frac{i}{4}\pi} \quad (8.9)$$

En effet, observons la simplification automatique du nombre $5\sqrt{2} e^{-\frac{1}{4}i\pi}$.

$$\begin{aligned} &> 5*\text{sqrt}(2)*\text{exp}(-(1/4*i)*\text{Pi}); \\ &5\sqrt{2} \left(\frac{\sqrt{2}}{2} - \frac{i\sqrt{2}}{2} \right) \end{aligned} \quad (8.10)$$

Par curiosité, voyons comment opère cette macro-commande maison.

$$\begin{aligned} &> \text{N:=polar}(r,\text{theta}): \\ &\quad \text{N:=convert}(N,\text{Exp}); \\ &\text{polar}(r,\theta) = |r| e^{i\arctan(r\sin(\theta), r\cos(\theta))} \end{aligned} \quad (8.11)$$

Pour terminer cette section, voyons comment contourner le mécanisme de la simplification automatique pour nous permettre d'énoncer dans une zone de requête la remarquable égalité:

$$\begin{aligned} &e^{i\pi} + 1 = 0 \\ &> \text{convert}(-1,\text{Exp})+1=\text{exp}(i*\text{Pi})+1; \\ &e^{i\pi} + 1 = 0 \end{aligned} \quad (8.12)$$

Notez, dans le membre de droite de la requête précédente, la simplification automatique de $e^{i\pi} + 1$ par 0.

Root et surd

Calculons la racine cubique de -8. La macro-commande `root(, n)` ou `root[n]()` donne la racine énième complexe principale.

$$\begin{aligned} &> \text{Racine:=root}[3](-8); \\ &\text{Racine} := 2(-1)^{1/3} \end{aligned} \quad (9.1)$$

Mais pourquoi donc que le mécanisme de la simplification automatique ne simplifie-t-il pas $(-1)^{1/3}$ par -1 ?

Sachez que la racine cubique principale de -1 n'est pas -1. (Voir la section *Représentation graphique des n racines énièmes d'un nombre complexe*)

$$\begin{aligned} &> \text{Racine=evalc}(\text{Racine}); \\ &2(-1)^{1/3} = 1 + i\sqrt{3} \end{aligned} \quad (9.2)$$

La macro-commande `surd(, n)` est à utiliser lorsqu'il s'agit des racines d'ordre n comme on les obtient habituellement dans les nombres réels.

$$\begin{aligned} &> \text{surd}(-1,3); \\ &\quad \text{surd}(-8,3); \\ &-1 \\ &-2 \end{aligned} \quad (9.3)$$

Lorsque l'ordre de la racine est pair (n pair) et que le nombre réel est négatif, `surd` donne quand même pour résultat une racine imaginaire.

$$\begin{aligned} &> \text{surd}(-5,2); \\ &i\sqrt{5} \end{aligned} \quad (9.4)$$

Avec Maple, la racine énième principale est obtenu en appliquant la formule

$$root(z, n) = e^{\frac{\ln(z)}{n}}$$

Le logarithme naturel de z , $\ln(z)$ est obtenu par la formule

$$\ln(z) = \ln(|z|) + \text{argument}(z) i \text{ où } -\pi < \text{argument}(z) \leq \pi$$

Ainsi,

$$\begin{aligned} e^{\frac{\ln(z)}{n}} &= e^{\frac{\ln(|z|)}{n} + \frac{\text{argument}(z) i}{n}} \\ &= e^{\ln\left(\frac{|z|}{n}\right)} e^{\frac{\text{argument}(z) i}{n}} \\ &= |z|^{\frac{1}{n}} e^{\frac{\text{argument}(z) i}{n}} \\ &= r^{\frac{1}{n}} e^{\frac{\text{argument}(z) i}{n}} \quad (\text{en écriture exponentielle d'un nombre complexe}) \end{aligned}$$

Le nombre $e^{\frac{\text{argument}(z) i}{n}}$ est donné par l'identité d'Euler $e^{i\theta} = \cos(\theta) + i \sin(\theta)$.

Bref, avec Maple, en sachant que $\theta = \text{argument}(z)$ n'est pas un angle nécessairement tel que $0 \leq \theta < 2\pi$, les nombres complexes situés dans les troisième et quatrième quadrant, leur racine énième principale donnée par l'évaluateur ne seront pas les mêmes que celles de la racine énième principale comme on pourrait l'avoir enseigné en classe.

En effet, en classe, on enseigne parfois à écrire un nombre complexe polaire avec un argument positif θ compris dans l'intervalle $[0, 2\pi[$ (forme trigonométrique). C'est ce qu'on appelle écrire le nombre complexe sous la forme trigonométrique. En conséquence, les n racines énièmes d'un nombre complexe seront notées par

$$z_k = r^{\frac{1}{n}} \text{cis}\left(\frac{\theta}{n} + \frac{2k\pi}{n}\right), k = 0 \dots n-1$$

Et on définit la racine énième principale (exprimée sous la forme trigonométrique) comme étant celle ayant une valeur d'argument minimal, soit par $z_0 = r^{\frac{1}{n}} \text{cis}\left(\frac{\theta}{n}\right)$.

Ainsi, pour $z = -\sqrt{2} \sqrt{2} i = 2 \text{cis}\left(\frac{5\pi}{4}\right)$, la racine cubique principale est $z_0 = 2^{\frac{1}{3}} \text{cis}\left(\frac{5\pi}{12}\right)$. Vérifions que

l'on a bien $z_0^3 = -\sqrt{2} \sqrt{2} i$.

```
> z[0]:=polar(2^(1/3),5*Pi/12);
evalc(z[0]^3);
```

$$z_0 := \text{polar}\left(2^{1/3}, \frac{5\pi}{12}\right)$$

$$-6 \sin\left(\frac{5\pi}{12}\right)^2 \cos\left(\frac{5\pi}{12}\right) + 2 \cos\left(\frac{5\pi}{12}\right)^3 + i \left(-2 \sin\left(\frac{5\pi}{12}\right)^3 + 6 \sin\left(\frac{5\pi}{12}\right) \cos\left(\frac{5\pi}{12}\right)^2 \right) \quad (9.5)$$

```
> simplify((9.5));
```

$$(-1-i)\sqrt{2} \quad (9.6)$$

Obtenons une approximation décimale de z_0 .

```
> z:=(polar(2^(1/3),5*Pi/12)):
z:=evalc(z);
evalf(``=rhs(%));
```

$$\text{polar}\left(2^{1/3}, \frac{5\pi}{12}\right) = 2^{1/3} \cos\left(\frac{5\pi}{12}\right) + i 2^{1/3} \sin\left(\frac{5\pi}{12}\right)$$

$$= 0.3260915631 + 1.216990281i \quad (9.7)$$

Vérifions maintenant si la racine principale donnée par Maple est bien une racine cubique de z et ensuite obtenons une approximation de la racine principale donnée par Maple.

```
> Racine:=root[3](-sqrt(2)-sqrt(2)*i):
Racine:=convert(Racine,polar);
evalc(``=Racine^3);
```

$$\left(-\sqrt{2} - i\sqrt{2}\right)^{1/3} = \text{polar}\left(2^{1/3}, -\frac{\pi}{4}\right)$$

$$= -\sqrt{2} - i\sqrt{2} \quad (9.8)$$

La racine obtenue avec la macro-commande `root[3]()` est bien une racine cubique complexe de $-\sqrt{2} - \sqrt{2}i$.

Obtenons une approximation de cette racine cubique.

```
> evalf(Racine);
```

$$0.8908987181 - 0.8908987181i \quad (9.9)$$

Ce n'est pas la même racine principale...car leur argument ne sont pas égaux à un multiple entier de 2π .

En effet, l'argument de la racine principale obtenue avec l'évaluateur est

```
> evalc(argument(Racine));
evalf(%);
```

$$-\frac{\pi}{4}$$

$$-0.7853981635 \quad (9.10)$$

tandis que l'argument calculé selon la définition donnée en classe de la racine principale est

```
> evalf(5*Pi/12);
```

$$1.308996939 \quad (9.11)$$

$2^{1/3} \text{cis}\left(\frac{5\pi}{12}\right)$ est l'une des racines qui sera obtenue en résolvant $z^3 = -\sqrt{2} - \sqrt{2}i$.

```
> z:='z':
Éq:=z^3=-sqrt(2)-sqrt(2)*i;
Racines:=evalf([solve(Éq,z)]);
```

$$\acute{E}q := z^3 = -\sqrt{2} - i\sqrt{2}$$

```
Racines := [0.8908987181 - 0.890898718i, -1.216990281 - 0.3260915635i, 0.3260915635
+ 1.216990282i]
```

```
> z:=polar(2^1/3,5*Pi/12):
z:=evalf(evalc(z));
```

$$\text{polar}\left(\frac{2}{3}, \frac{5\pi}{12}\right) = 0.1725460301 + 0.6439505509i \quad (9.13)$$


Analyse réelle et calculs dans C

Pour tracer le graphique de la fonction racine cubique ($y = \sqrt[3]{x}$), le néophyte utilisera probablement la macro-commande `root[3]()`.

```
> setoptions(labels=["x","y"],font=[axes,TIMES,8],labeldirections=
[horizontal,horizontal]);
```

```
> f:=x->root[3](x);
plot([x,f(x),x=-4..4],view=[-2..2,-2..2]);
```

$$f := x \mapsto \text{root}_3(x)$$




Comme on le voit, ce n'est pas le résultat attendu. En tout temps, et donc aussi pour le tracé de graphique, Maple a comme référentiel les nombres complexes. Et, comme vous savez maintenant que la racine cubique principale d'un nombre négatif est un nombre imaginaire, c'est ce qui explique qu'il n'y a aucun tracé dans la partie négative de l'abscisse.

Vous devez donc créer la fonction racine cubique avec la macro-commande `surd(x,3)`.

```
> f:=x->surd(x,3);
plot([x,f(x),x=-2..2],view=[-2..2,-2..2]);
```

$$f := x \mapsto \sqrt[3]{x}$$



Voyons maintenant un développement qui saura, une fois pour toute, vous convaincre à ne pas utiliser Maple comme une boîte noire. Il y a un réel apprentissage de Maple qu'il faut faire.

Soit les fonctions f et g définies par $f(x) = \sqrt{\frac{x^2-4}{x^2-9}}$ et $g(x) = \frac{\sqrt{x^2-4}}{\sqrt{x^2-9}}$.

Obtenez les graphiques de f et g .


```
> f:=x->sqrt((x^2-4)/(x^2-9));
g:=x->sqrt(x^2-4)/sqrt(x^2-9);
```

$$f := x \mapsto \sqrt{\frac{x^2-4}{x^2-9}}$$

$$g := x \mapsto \frac{\sqrt{x^2-4}}{\sqrt{x^2-9}}$$

(10.1)

```
> A:=Array(1..2):
A[1]:=plot([x,f(x),x=-8..8],titlefont=["ROMAN", 10],title=(y=f(x))):
A[2]:=plot([x,g(x),x=-8..8],titlefont=["ROMAN", 10],title=(y=g(x))):
display(A,scaling=unconstrained ,size=[300,300]);
```



Sur la base de leur tracé respectif, nous devons conclure que les fonctions f et g sont identiques. Nous allons montrer que c'est évidemment faux.

Recherchons le domaine de la fonction f.

```
> solve((x^2-4)/(x^2-9)>=0,{x});
```

$$\{x < -3\}, \{-2 \leq x, x \leq 2\}, \{3 < x\}$$
(10.2)

Recherchons le domaine de la fonction g.

```
> solve({(x^2-4)>=0,(x^2-9)>0},{x});
```

$$\{x < -3\}, \{3 < x\}$$
(10.3)

Le tracé de la fonction g est évidemment erroné. Comment est-ce possible ?

Qu'en est-il de celui de la fonction f ?

La macro-commande `plot`, au moment de la création de la structure graphique, fait appel à l'évaluateur qui calcule les coordonnées des points à tracer. Et, comme vous le savez maintenant, ce calcul est effectué dans l'ensemble des nombres complexes. Pour $-2 \leq x \leq 2$, le calcul des images, selon la règle g, a posé, à chaque fois, un quotient de deux imaginaires (purs) dont la simplification a donné un nombre réel.

Par exemple, détaillons le calcul de g(1). Obtenons d'abord les valeurs des numérateur et dénominateur.

```
> g(x);
```

```
  a:=eval(numer(g(x)),x=1);
```

```
  b:=eval(denom(g(x)),x=1);
```

$$\frac{\sqrt{x^2-4}}{\sqrt{x^2-9}}$$

$$a := \sqrt{-3}$$

$$b := \sqrt{-8}$$
(10.4)

Ensuite, obtenons le quotient.

```
> Quotient:=a/b; # rationalisation automatique
```

```
  Quotient=combine(Quotient,radical);
```

```
  evalf(``=rhs(%));
```

$$Quotient := -\frac{\sqrt{-3}\sqrt{-8}}{8}$$

$$-\frac{\sqrt{-3}\sqrt{-8}}{8} = \frac{\sqrt{6}}{4}$$

$$= 0.6123724358$$
(10.5)

Puisque le référentiel pour chaque tracé est pour nous l'ensemble des nombres réels \mathbb{R} et que pour Maple, c'est l'ensemble des nombres complexes \mathbb{C} , **il est donc très important** de prendre le temps **de trouver le domaine d'une fonction (ou d'une équation à résoudre)** avant de faire quelque analyse que ce soit dans \mathbb{R} .

Puisque la division ou la multiplication peut poser problème dans le tracé d'une fonction réelle à valeur réelle, on pourrait très bien créer, en Maple, un opérateur qui pourrait prendre en charge la simplification de la division (ou de la multiplication) impliquant des nombres imaginaires. Après tout, Maple est ce qu'on appelle *un système ouvert*. L'opérateur que nous allons créer retournera un nombre réel à la condition qu'il soit le

résultat de la division de deux nombres réels. Les deux nombres devront être du type `numeric` : soit du type `integer`, `fraction`, ou `float`. Cet opérateur bloquera donc, entre autres, tous les résultats réels de la division de deux nombres imaginaires.

À l'aide de la macro-commande `define`, donnons le nom `DIV` à ce nouvel opérateur.

```
> a:='a': # Il faut que a et b soient libres
    b:='b':
    define(DIV,DIV(a::numeric,b::numeric)=a/b);
```

Il n'y a aucun "sortie". Comme, par exemple, avec les macro-commandes `restart` et `assign`.

Observons comment `DIV` opère la "nouvelle" division: si au moins un des deux opérandes n'est pas du type `numeric`, la division ne s'effectuera pas.

```
> DIV(-5,sqrt(2)); # sqrt(2) est du type ^ et non pas du type
    numeric
    DIV(-5,sqrt(2.)); # sqrt(2.) est du type float et donc du type
    numeric
    DIV(-5,2);
    DIV(sqrt(3.)*i,sqrt(8.)*i);
```

$$DIV(-5, \sqrt{2})$$

$$-3.535533907$$

$$-\frac{5}{2}$$

$$DIV(1.732050808i, 2.828427125i)$$

(10.6)

Maintenant, créons une nouvelle "fonction" `G` en opérant la division des expressions $\sqrt{x^2 - 4}$ et $\sqrt{x^2 - 9}$ avec l'opérateur `DIV`.

```
> G:=x->DIV(sqrt(x^2-4),sqrt(x^2-9));
    G := x ↦ DIV(√(x²-4), √(x²-9))
```

(10.7)


Dans la macro-commande `plot`, à chaque assignation à x (abscisse) **d'un nombre décimal** de l'intervalle de traçage, $G(x)$ (ordonnée) pointerait vers un nombre réel à la condition d'être le résultat de l'opérateur `DIV`.

```
> G(1.5);
    G(5.);
    DIV(1.322875656i, 2.598076211i)
    1.145643924
```

(10.8)

Traçons maintenant la fonction `G` dans les mêmes conditions de l'ancienne fonction `g` mais ajoutez en option `view=[-6..6., 0..6]` pour contrôler l'affichage de l'axe des y .

```
> plot([x,G(x),x=-6..6],numpoints=300,scaling=unconstrained,
    titlefont=[TIMES,ITALIC,10],view=[-6..6.,0..6]);
```



pour terminer, voici un dernier développement qui va encore illustrer une situation de simplification de nombres imaginaires donnant un nombre réel.

Soit à résoudre l'équation $\frac{4}{x + \sqrt{2 - x^2}} + \frac{4}{x - \sqrt{2 - x^2}} = 2x$.

```
> Problème:=4/(x+sqrt(2-x^2))+4/(x-sqrt(2-x^2))-2*x;
Racines:=solve(Problème,{x});
```

$$\text{Problème} := \frac{4}{x + \sqrt{-x^2 + 2}} + \frac{4}{x - \sqrt{-x^2 + 2}} = 2x$$

$$\text{Racines} := \{x=0\}, \{x=\sqrt{3}\}, \{x=-\sqrt{3}\}$$

(10.9)

L'évaluateur, en résolvant dans \mathbb{C} , a trouvé trois racines réelles (complexes aussi) à cette équation (n'oublions pas qu'un nombre réel est un nombre complexe qui s'ignore:). Si vous aviez pris soin de préciser, dès le départ, le domaine de définition de cette équation dans \mathbb{R} , la seule racine réelle dont vous seriez tenu de contrôler analytiquement est $x = 0$.

En effet, montrons qu'avec $x = \sqrt{3}$ (l'une des racines trouvées), le membre de gauche de l'équation à résoudre a une simplification dans \mathbb{C} qui donne un nombre réel.

```
> eval(Problème,Racines[2]);
radnormal(%);
```

$$\frac{4}{\sqrt{3} + i} + \frac{4}{\sqrt{3} - i} = 2\sqrt{3}$$

$$2\sqrt{3} = 2\sqrt{3}$$

(10.10)


Il est possible d'imposer à l'évaluateur la résolution d'équation dans \mathbb{R} lorsque nous utilisons la macro-commande solve().

```
> use RealDomain in solve(Problème,{x}) end use
{x=0}
```

(10.11)

Malheureusement, l'environnement RealDomain ne gère pas dans \mathbb{R} la macro-commande plot().

```
> use RealDomain in plot([x,g(x),x=-8..8],view=[-8..8,-2..6]) end use
```



Représentation graphique des n racines énièmes d'un nombre complexe

Définition de la macro-commande `Racines_Complexes`

La macro-commande

`Racines_Complexes(Nombre, Ordre, Avec_Points)`

possède trois arguments:

- **Nombre** étant le nombre complexe dont on recherche les n racines énièmes
- **Ordre** étant l'ordre de la racine. Ordre doit être un entier positif
- **Avec_Points** permettant un contrôle de l'affichage des racines: oui et non (en fait, tout autre chaîne de caractère que oui: n par exemple).

La macro-commande **`Racines_Complexes`** illustrera les n racines énièmes complexes d'un nombre complexe donné avec une écriture rectangulaire ou avec une écriture polaire.

Cette illustration sera faite à l'aide de vecteurs pointant vers les sommets d'un polygone régulier inscrit dans un cercle (si $\text{Ordre} > 2$). Dans le graphique, la racine principale sera illustré par un vecteur rouge.

Initialisation

Pour activer la macro-commande **`Racines_Complexes`**, exécutons le bloc d'instructions ci-dessous.

```
> restart;
> Racines_Complexes:=proc(C::complex,O::posint,
  Avec_Points::symbol)
# -----
# Définition des variables locales
  local c,k,arêtes,Détails,Équation,Flèches,Flèches_P,
    Liste,ordre_convexe,Points,points,Racines,
    Racines_intermédiaires,Racine_principale,Sol_générale,t,
  z,Zéros:
# -----
# Initialisation
  interface(imaginaryunit=i);
```

```

Détails:=Avec_Points;
_EnvAllSolutions := true:
plots[setoptions](size=[300,300],xtickmarks=12,ytickmarks=12,
    labels=["Re","Im"],
    font=[axes,"TIMES",8],
    labeldirections=["horizontal","vertical"],
    scaling=constrained);
interface(typesetting=extended);
Typesetting:-Settings(striptrailing=true);
# -----
# Critère de tri
    ordre_convexe:=(x,y)->evalb(evalf(op(2,y)>0) and evalb(evalf
(op(1,x)<op(1,y)))):
# -----
# Calcul des racines
Équation:=z^O=C;
Sol_générale:=evalc(solve(z^n=C,z)):
if type(Sol_générale,RootOf) then
    Racines:=evalc([allvalues(subs(n=O,Sol_générale))]);
    Points:=sort([seq([Re(Racines[k]),Im(eval(Racines[k]))],k=1..
O)],ordre_convexe):
else
    t:=op(1,indets(Sol_générale)):
    Racines_intermédiaires:=subs(t=k,Sol_générale);
    Racines:=subs(n=O,Racines_intermédiaires);
    Racine_principale:=root[O](C);
    Zéros:=seq(subs(k=t,Racines),t=0..O-1);
    Points:=sort([seq([Re(evalf(Zéros[k])),Im(evalf(Zéros[k]))],
`k`=1..nops(Zéros))],ordre_convexe):
    fi;
# -----
# Affichage conditionnel des résultats
if Détails=oui then
    print(Équation);
    print(`La racine principale d'ordre `,O,`de`,C,`est `,
simplify(evalc(root[O](C))));
    print(` ou soit numériquement `,evalf(evalc(root[O](C))));
fi;
# -----
# Affichage graphique des racines énièmes
arêtes:=plots[polygonplot](Points,linestyle=2,color=white,
transparency=0,thickness=2):
points:=plots[polygonplot](evalf(Points),style=point,symbol=
solidcircle,symbolsize=18,color=navy):
Flèches_P:=plottools[arrow]([0,0],Points[O], .04, .08, .15,

```

```

color=red):
    Flèches:=seq(plottools[arrow]([0,0],Points[k+1], .04, .08,
    .15, color=navy),k=0..0-2):
    c:= plottools[circle]([0,0], sqrt(Points[1,1]^2+Points[1,2]
    ^2), linestyle=4,color=grey):
    plots[display](Flèches_P,Flèches,c,arêtes,points):
end proc:

```


Illustration des deux racines deuxièmes de -1

```
> Racines_Complexes(-1,2,oui);
```

$$z^2 = -1$$

La racine principale d'ordre ,2, de, -1, est, i

ou soit numériquement ,i



```
> Racines_Complexes(-1,2,non);
```





Illustration des trois racines troisièmes de -1

```
> Racines_Complexes(-1,3,oui);
```

$$z^3 = -1$$

La racine principale d'ordre ,3, de, -1, est, $\frac{1}{2} + \frac{i\sqrt{3}}{2}$

ou soit numériquement , 0.5 + 0.866025404 i



Remarque: La racine cubique principale du nombre réel -1 n'est pas réelle: elle est imaginaire.

Illustration des quatres racines quatrièmes de $\text{polar}\left(3, \frac{5\pi}{3}\right)$

```
> Racines_Complexes(polar(3,5*Pi/3),4,oui);
Error, invalid input: Racines_Complexes expects its 1st
argument, C, to be of type complex, but received polar(3,
(5/3)*Pi)
```


La procédure maison `Racines_Complexes()` teste d'abord que le nombre passé en procédure soit du type `complex`. Bien que $\text{polar}\left(3, \frac{5\pi}{3}\right)$ soit syntaxiquement correcte, pour Maple ce n'est pas un objet de type `complex`. Faisons donc préalablement une conversion avec `evalc`.

```
> Racines_Complexes(evalc(polar(3,5*Pi/3)),4,oui);
```

$$z^4 = \frac{3}{2} - \frac{3i\sqrt{3}}{2}$$

La racine principale d'ordre 4, de $\frac{3}{2} - \frac{3i\sqrt{3}}{2}$, est, $-\frac{\sqrt{2}(-1-i+(-1+i)\sqrt{3})}{4} 3^{1/4}$

ou soit numériquement, $1.271229878 - 0.3406250193i$



Puisque $\text{polar}\left(3, \frac{5\pi}{3}\right) = \text{polar}\left(3, -\frac{\pi}{3}\right)$, on obtiendra le même résultat que précédemment.

```
> Racines_Complexes(evalc(polar(3,-Pi/3)),4,oui);
```

$$z^4 = \frac{3}{2} - \frac{3i\sqrt{3}}{2}$$

La racine principale d'ordre , 4, de, $\frac{3}{2} - \frac{3i\sqrt{3}}{2}$, est, $-\frac{\sqrt{2}(-1-i+(-1+i)\sqrt{3})}{4} 3^{1/4}$
ou soit numériquement , $1.271229878 - 0.3406250193i$




Illustration des cinq racines cinquièmes de $32i$

> `Racines_Complexes(32*i,5,oui);`

$$z^5 = 32i$$

La racine principale d'ordre , 5, de, $32i$, est, $2i \cos\left(\frac{2\pi}{5}\right) + 2 \sin\left(\frac{2\pi}{5}\right)$

ou soit numériquement , $1.902113033 + 0.6180339888i$





Illustration des onze racines onzièmes de 1

> `Racines_Complexes(1,11,oui);`

$$z^{11} = 1$$

La racine principale d'ordre , 11, de, 1, est, 1

ou soit numériquement , 1.



Approche empirique dans le plan d'Argand de la $\lim_{n \rightarrow \infty} (-1)^{\frac{1}{n}}$


Pour réaliser l'animation ci-dessous, il est requis de ne pas effectuer l'affichage des résultats successifs que donnera la macro-commande **Racines_Complexes**. Pour cela, il est fortement recommandé de donner au troisième argument la valeur non. Si vous précisez la valeur oui, il y aura un long affichage des racines n èmes pour chacun des termes de la séquence, ce qui n'est pas souhaitable ici.

Il faut être un tantinet patient car la construction de 50 tracés prendra plusieurs secondes (selon votre machine)

```
> séquences:=seq(Racines_Complexes(-1,a,non),a=1..50):
```

Après l'exécution de la requête ci-dessous, cliquez le dessin pour le sélectionner afin de faire apparaître le menu contextuel. Cliquez ensuite le bouton de départ pour démarrer l'animation.

```
> plots[display](séquences,insequence=true);
```



Obtenons directement cette limite avec la macro-commande `limit()`.

```
> Limit((-1)^(1/n),n=infinity)=limit((-1)^(1/n),n=infinity);
```

$$\lim_{n \rightarrow \infty} (-1)^{\frac{1}{n}} = 1 \quad (12.1)$$

Exercices

No. 1

Effectuer

a) $\frac{i}{1-i}$ b) $\frac{2+i}{i}$ c) $\frac{-5i}{2+i} + 2 + 3i$ d) $\left(\frac{1}{2} - \frac{\sqrt{3}i}{2}\right)\left(\frac{\sqrt{3}}{2} + \frac{i}{2}\right)$

No. 2Soit $z_1 = 1 + 3i$ et $z_2 = 3 - i$. Calculer

a) $\begin{bmatrix} z_1 \\ z_2 \end{bmatrix}$ b) $\frac{\bar{z}_1}{z_2}$

No. 3Si $z_1 = 1 + 4i$ et $z_2 = 2 - 3i$. Déterminer

a) $\Re(z_1 z_2)$ b) $\Re\left(\frac{z_1}{z_2}\right)$ c) $\Im(z_1 z_2)$ d) $\Im\left(\frac{z_1}{z_2}\right)$

No. 4Montrer en utilisant les propriétés des conjugués, que $\forall z \in \mathbb{C}$,

a) $\overline{\bar{z} + i} = z - i$ b) $\overline{iz} = -i\bar{z}$ c) $\frac{z}{i} = i\bar{z}$

No. 5Résoudre dans \mathbb{C} l'équation $\frac{(1-i)z}{2+i} = \frac{(1+i)z+4}{1-2i}$ **No. 6**Montrer que si $z = a + bi$ ($a, b \in \mathbb{R}$) satisfait l'équation $z^2 + z + 1 = 0$ alors \bar{z} satisfait aussi cette équation.**No. 7**Représenter, dans le plan complexe, le lieu géométrique de tous les points d'affixes $z \in \mathbb{C}$ satisfaisant l'équation:

a) $\Re(z) = -\Im(z)$ b) $\Re(z) = 2$ c) $\Im(z) = -1$ d) $z = \bar{z}$ e) $|z-3| = 4$ f) $|z-5i| \leq 3$ g) $\Re(z)^2 = \Im(z)^2$ h) $2 < |z+2-i| \leq 4$ i) $|z-\bar{z}| = 1$ j) $1 < |z-1-i| \leq 2$

No. 8À l'aide uniquement des propriétés des modules, montrer que $|z| = |\bar{z}| = |-z| = |-iz| = \left|\frac{z}{i}\right|$.**No. 9**Calculer le module de z si

a) $z = \frac{5 + 11\sqrt{3}i}{7 - 4\sqrt{3}i}$ b) $z = \frac{13 + 5i}{13 - 5i}$ c) $z = 648 - 648\sqrt{3}i$.

No. 10Montrer que $|z-w| = |w-z|$, $\forall z, w \in \mathbb{C}$.**No. 11**

Écrire sous la forme polaire (Maple) les nombres complexes suivants.

a) $z = -6 + 6\sqrt{3}i$ b) $z = -1 - \sqrt{3}i$ c) $z = \pi$ d) $z = -7 + 24i$ e) $z = \frac{4}{5} + \frac{3}{5}i$

No. 12Écrire sous la forme exponentielle les nombres complexes du No.11. Utiliser `convert(, Exp)`.**No. 13**

Écrire sous la forme rectangulaire les nombres complexes suivants.

a) $z = \text{cis}(0)$ b) $z = \sqrt{6} \text{cis}\left(\frac{\pi}{4}\right)$ c) $z = \sqrt{2} \text{cis}(30^\circ)$ d) $z = 2 \text{cis}(315^\circ)$

No. 14

Écrire sous la forme exponentielle les nombres complexes du No.13. Utiliser `convert(, Exp)`.

No. 15

Effectuer directement les calculs sous la forme polaire. Utilisez la macro-commande `simplify()`.

a) $\text{polar}\left(3, \frac{\pi}{4}\right) \text{polar}\left(2, \frac{\pi}{3}\right)$ b) $\text{polar}\left(-4, \frac{\pi}{4}\right) \text{polar}\left(2, \frac{\pi}{3}\right)$ c) $\text{polar}\left(2, \frac{\pi}{6}\right)^5$ d)
 $\text{polar}\left(2, \frac{\pi}{6}\right)^{-5}$

No. 16

Représenter graphiquement les trois racines troisièmes de

a) -27 b) $-27i$ c) $-27(-i)$

Quel effet, la multiplication de -27 par i , observez-vous sur les trois racines cubiques de -27 ? Et qu'en est-il par la multiplication de -27 par $-i$?

No. 17

Déterminer la somme et le produit des six racines sixièmes de 1.

No. 18

Soit la fonction f définie par $f(x) = \sqrt{x^2 - 4} \sqrt{x^2 - 9}$

a) Dans \mathbb{C} , résoudre simultanément les inéquations $x^2 - 4 \geq 0$ et $x^2 - 9 \geq 0$.

b) Même question qu'en a) mais la résolution dans \mathbb{R} .

c) Créer un opérateur MUL bloquant le mécanisme de la simplification automatique dans \mathbb{C} de la multiplication de deux nombres imaginaires dont le résultat serait un nombre réel. Par exemple

$$\sqrt{-2} \sqrt{-2} = -2.$$

d) Obtenir dans un premier temps le graphique de la fonction f sur l'intervalle $]-4, 4]$ en laissant intervenir le mécanisme de la simplification automatique dans \mathbb{C} de Maple pour le produit $\sqrt{x^2 - 4} \sqrt{x^2 - 9}$. Spécifier l'option `view=[-4..4, -8..6]`.

e) Obtenir ensuite le graphique de la fonction f sur l'intervalle $]-4, 4]$ en bloquant le mécanisme de la simplification automatique dans \mathbb{C} de Maple à l'aide de l'opérateur MUL. Spécifier l'option `view=[-4..4, -8..6]`.